
Change 1 function f_GERAN_GPRS_Authentication ()
	Function name
	f_GERAN_GPRS_Authentication ()

	Reason for change
	In the current TTCN implementation, after the assignment call  the function call v_SecurityInfo.authPS := f_AuthenticationInit(v_SecurityInfo.authPS);, the call to function f_GERAN_SecurityInfo_Set(v_SecurityInfo); is not made.
Thus, updated the securityInfo is not stored.
This results in the function f_LLC_TLLI_Assign(f_GERAN_SecurityInfo_Get() returning the non-updated securityInfo, which can cause issues later in the testcase(s)
This needs to be addressed.

	Summary of change
	Called the function f_GERAN_SecurityInfo_Set after the assignment call v_SecurityInfo.authPS := f_AuthenticationInit(v_SecurityInfo.authPS);
Please see below.

	TTCN module
	Common/GERAN/GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before Change:
	function f_GERAN_GPRS_Authentication(GERAN_CellId_Type p_CellId) runs on GERAN_PTC
  {
    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;
    var TLLI v_TLLI;
    var GERAN_SecurityInfo_Type v_SecurityInfo := f_GERAN_SecurityInfo_Get();
    var B32_Type    v_AuthRES_High;
    var bitstring   v_AuthRES;
    var GERAN_CellId_Type v_LLMEId := p_CellId; // @sic R5s120572 sic@
    v_SecurityInfo.authPS := f_AuthenticationInit(v_SecurityInfo.authPS);
    // Send AUTHENTICATION AND CIPHERING REQUEST using relevant PS keys computed before.
    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req(f_GERAN_TLLI_Get(p_CellId),
                                                               tsc_LLCSapi_GMM,
                                                               tsc_LLC_PM,
                                                               tsc_LLC_NoCiph,
                                                               cs_G_AuthAndCiphReq(cs_GMM_AuthRAND(v_SecurityInfo.authPS.RandValue),
                                                                                   cs_GMM_KeySeq(v_SecurityInfo.authPS.KeySeq),
                                                                                   v_SecurityInfo.gprsCipherAlg,
                                                                                   cs_GMM_AUTN(v_SecurityInfo.authPS.AUTN)),
                                                               v_LLMEId)); // @sic R5s120572 sic@
    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg(p_CellId, car_G_LLC_UnitData_IND
                                            ((cr_G_AuthAndCiphRsp(cr_AuthRspAny, cr_AuthCiphRspExtAny),
                                              cr_G_AuthAndCiphRsp(cr_AuthRspAny, omit),
                                              cr_G_AuthAndCiphFailureAny)));
    if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphRsp(cr_AuthRspAny, cr_AuthCiphRspExtAny))))
      {// AUTHENTICATION AND CIPHERING RESPONSE including both Authentication Response parameters (RES and RES ext)
        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;
        v_AuthRES := v_AuthRES_High & v_LLCDataInd.unit.msg.authCipherResp.authRspExt.rES;
        // don't change the TLLI, only the ciphering algorithm
        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF
        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);
        // Verify that the received Authentication Response parameters match expected response:
        if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES))) {
          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");
        }
        if (lengthof(v_AuthRES) != 0) { // @sic R5s190073 sic@
          v_SecurityInfo.authPS.XRESLength := lengthof(v_AuthRES) / 8;
        }
      }
    else if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphRsp(cr_AuthRspAny, omit))))
      {// AUTHENTICATION AND CIPHERING RESPONSE including only one Authentication Response parameter (RES)
        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;
        // don't change the TLLI, only the ciphering algorithm
        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF
        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);
        // Verify that the received Authentication Response parameters match expected response.
        if (v_AuthRES_High != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES_High))) {
          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");
        }
        if (lengthof(v_AuthRES_High) != 0) { // @sic R5s190073 sic@
          v_SecurityInfo.authPS.XRESLength := lengthof(v_AuthRES_High) / 8;
        }
      }
    else if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphFailureAny)))
      { //AUTHENTICATION AND CIPHERING Failure, so stop execution
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Received Authentication and Ciphering Failure");
      }
    else { // shouldn't get here because it should match with one of the above 3
      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected message");
    }
    f_GERAN_SecurityInfo_Set(v_SecurityInfo);  // @sic R5s110596, R5s190073 sic@
  }//end of f_GPRS_Authentication


After Change:

	function f_GERAN_GPRS_Authentication(GERAN_CellId_Type p_CellId) runs on GERAN_PTC
  {
    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;
    var TLLI v_TLLI;
    var GERAN_SecurityInfo_Type v_SecurityInfo := f_GERAN_SecurityInfo_Get();
    var B32_Type    v_AuthRES_High;
    var bitstring   v_AuthRES;
    var GERAN_CellId_Type v_LLMEId := p_CellId; // @sic R5s120572 sic@
    v_SecurityInfo.authPS := f_AuthenticationInit(v_SecurityInfo.authPS);
    f_GERAN_SecurityInfo_Set(v_SecurityInfo);  
    // Send AUTHENTICATION AND CIPHERING REQUEST using relevant PS keys computed before.
    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req(f_GERAN_TLLI_Get(p_CellId),
                                                               tsc_LLCSapi_GMM,
                                                               tsc_LLC_PM,
                                                               tsc_LLC_NoCiph,
                                                               cs_G_AuthAndCiphReq(cs_GMM_AuthRAND(v_SecurityInfo.authPS.RandValue),
                                                                                   cs_GMM_KeySeq(v_SecurityInfo.authPS.KeySeq),
                                                                                   v_SecurityInfo.gprsCipherAlg,
                                                                                   cs_GMM_AUTN(v_SecurityInfo.authPS.AUTN)),
                                                               v_LLMEId)); // @sic R5s120572 sic@
    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg(p_CellId, car_G_LLC_UnitData_IND
                                            ((cr_G_AuthAndCiphRsp(cr_AuthRspAny, cr_AuthCiphRspExtAny),
                                              cr_G_AuthAndCiphRsp(cr_AuthRspAny, omit),
                                              cr_G_AuthAndCiphFailureAny)));
    if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphRsp(cr_AuthRspAny, cr_AuthCiphRspExtAny))))
      {// AUTHENTICATION AND CIPHERING RESPONSE including both Authentication Response parameters (RES and RES ext)
        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;
        v_AuthRES := v_AuthRES_High & v_LLCDataInd.unit.msg.authCipherResp.authRspExt.rES;
        // don't change the TLLI, only the ciphering algorithm
        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF
        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);
        // Verify that the received Authentication Response parameters match expected response:
        if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES))) {
          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");
        }
        if (lengthof(v_AuthRES) != 0) { // @sic R5s190073 sic@
          v_SecurityInfo.authPS.XRESLength := lengthof(v_AuthRES) / 8;
        }
      }
    else if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphRsp(cr_AuthRspAny, omit))))
      {// AUTHENTICATION AND CIPHERING RESPONSE including only one Authentication Response parameter (RES)
        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;
        // don't change the TLLI, only the ciphering algorithm
        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF
        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);
        // Verify that the received Authentication Response parameters match expected response.
        if (v_AuthRES_High != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES_High))) {
          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");
        }
        if (lengthof(v_AuthRES_High) != 0) { // @sic R5s190073 sic@
          v_SecurityInfo.authPS.XRESLength := lengthof(v_AuthRES_High) / 8;
        }
      }
    else if (match(v_LLCDataInd, car_G_LLC_UnitData_IND(cr_G_AuthAndCiphFailureAny)))
      { //AUTHENTICATION AND CIPHERING Failure, so stop execution
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Received Authentication and Ciphering Failure");
      }
    else { // shouldn't get here because it should match with one of the above 3
      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected message");
    }
    f_GERAN_SecurityInfo_Set(v_SecurityInfo);  // @sic R5s110596, R5s190073 sic@
  }//end of f_GPRS_Authentication


